

Bayesian Biostatistics Using BUGS



**Βιο-Στατιστική κατά Bayes
με τη χρήση του Λογισμικού
BUGS**

ΜΑΘΗΜΑ 3: ΣΥΝΘΕΤΑ ΜΟΝΤΕΛΑ

Ioannis Ntzoufras

E-mail: ntzoufras@aueb.gr

Department of Statistics,
Athens University of
Economics & Business



5. ΣΥΝΘΕΤΑ ΜΟΝΤΕΛΑ ΣΤΟ BUGS

ΜΑΘΗΜΑ 3 (ΠΕΡΙΕΧΟΜΕΝΑ):

- #5.1 Prior Distributions
- #5.2 Parameterization
- #5.3 Random Effects
- #5.4 Examples
 - 5.4.1 Rats (Repeated Measures Regression)
 - 5.4.2 Seeds (Random Effects Logistic Regression)
 - 5.4.3 Surgical (Institutional Ranking)
 - 5.4.4 Equiv (Bioequivalence in Cross-over Trials)

5. ΣΥΝΘΕΤΑ ΜΟΝΤΕΛΑ ΣΤΟ BUGS

ΜΑΘΗΜΑ 3 (ΠΕΡΙΕΧΟΜΕΝΑ):

- ⌘ 5.1 Prior Distributions
- ⌘ 5.2 Parameterization
- ⌘ 5.3 Random Effects
- ⌘ 5.4 Examples
 - ◻ 5.4.5 Dyes (Variance Component Models)
 - ◻ 5.4.6 Blocker (Meta analysis & random Effects)
 - ◻ 5.4.7 Mice (Weibull Survival Analysis)
 - ◻ 5.4.8 Alli(Multinomial-Logistic Models)

5.1. ΚΑΘΟΡΙΣΜΟΣ PRIOR ΚΑΤΑΝΟΜΩΝ

- ⌘ Εάν έχουμε εκ-των-προτέρων πληροφορία από άλλες μελέτες τότε μπορούμε να χρησιμοποιήσουμε:
 - ◻ Κανονική για ποσοτικές (μέσος=prior belief, διακύμανση = αβεβαιότητα για την prior γνώση)
 - ◻ Γάμμα ή Λογαριθμο-κανονική για ποσοτικές θετικές
 - ◻ Βήτα για ποσοστά ή παραμέτρους στο [0,1]

5.1. ΚΑΘΟΡΙΣΜΟΣ PRIOR ΚΑΤΑΝΟΜΩΝ

Μη πληροφοριακές εκ-των-προτέρων κατανομές (non-informative prior distributions) :

- ⌘ δεν επηρεάζουν την συμπερασματολογία μας
- ⌘ βασιζόμαστε μόνο στα δεδομένα
- ⌘ εκ-των-υστερων αποτελέσματα είναι παρόμοια με της μέγιστης πιθανοφάνειας

5.1. ΚΑΘΟΡΙΣΜΟΣ PRIOR ΚΑΤΑΝΟΜΩΝ

Μη πληροφοριακές εκ-των-προτέρων κατανομές :

- ⌘ δίνουν ίδια πιθανότητα εμφάνισης μίας τιμής ή ενός ισομηκους διαστήματος (ομοιόμορφη κατανομή).
- ⌘ Η ομοιόμορφη ορίζεται σε ένα διαστήμα $[a, \beta]$.
- ⌘ Για μεταβλητές στο $R \Rightarrow U(-1000, 1000)$
- ⌘ Για μεταβλητές στο $[0, +\infty) \Rightarrow U(0, 1000)$
- ⌘ Για μεταβλητές στο $[0, 1] \Rightarrow U(0, 1)$
- ⌘ **KΙΝΔΥΝΟΣ:** Να χασουμε τιμές λόγω των ορίων a, β

5.1. ΚΑΘΟΡΙΣΜΟΣ PRIOR ΚΑΤΑΝΟΜΩΝ

Μη πληροφοριακές εκ-των-προτέρων κατανομές:

- ⌘ Εναλλακτικά μπορούμε να χρησιμοποιήσουμε improper prior κατανομές $f(x) \propto 1/c$ (σταθερά)
- ⌘ Στο BUGS προτείνουμε
 - ☒ Για μεταβλητές στο $R \Rightarrow$ Κανονική με μεγάλη διακύμανση (μικρή ακρίβεια)
 - ☒ Για μεταβλητές στο $[0, +\infty)$ \Rightarrow Γάμμα με $\alpha = \beta$ (μικρές τιμές)
[μέση τιμή 1, διάκυμανση μεγάλη = $1/\beta$].

5.2. ΠΑΡΑΜΕΤΡΟΠΟΙΗΣΗ ΤΟΥ ΜΟΝΤΕΛΟΥ

ΠΑΡΑΜΕΤΡΟΠΟΙΗΣΗ:

- ⌘ Σημαντική για τον υπολογισμό των παραμέτρων
- ⌘ Μια παράμετρος δεν είναι προσδιορίσιμη (identifiable) όταν τα δεδομένα δεν δίνουν αρκετή πληροφορία για τον υπολογισμό της.
- ⌘ Για το λόγο αυτό θέτουμε περιορισμούς:
 - ☒ Περιορισμός Αθροίσματος στο μηδέν (Sum-to-zero constraint)
 - ☒ Γωνιακός Περιορισμός (corner constraint) [treatment effect]

5.2. ΠΑΡΑΜΕΤΡΟΠΟΙΗΣΗ ΤΟΥ ΜΟΝΤΕΛΟΥ

ΠΑΡΑΜΕΤΡΟΠΟΙΗΣΗ:

- ⌘ Αν έχουμε το ίδιο μοντέλο με διαφορετική παραμετροποίηση τότε
 - ─ Η ερμηνεία των παραμέτρων είναι διαφορετική
 - ─ Οι προσαρμοσμένες ή προβλεπόμενες τιμές είναι ίδιες
 - ─ Οι παράμετροι του ίδιου μοντέλου μπορούν να μετασχηματιστούν σε παραμέτρους με άλλους περιορισμούς με απλές συναρτησεις
- ⌘ Η παραμετροποίηση αλλάζει και χρησιμοποιώντας διαφορετικές ψευδομεταβλητές (dummy variables).

5.2. ΠΑΡΑΜΕΤΡΟΠΟΙΗΣΗ ΤΟΥ ΜΟΝΤΕΛΟΥ

ΠΑΡΑΔΕΙΓΜΑ 1: One-way Anova

Έχουμε μία ποσοτική Y και μία κατηγορική A με κ επίπεδα

$$Y_i \sim \text{Normal}(\mu_i, \sigma^2)$$

- ⌘ **Παραμετροποίηση 1:** Εκτιμούμε κατευθείαν τους μέσους ανά ομάδα

5.2. ΠΑΡΑΜΕΤΡΟΠΟΙΗΣΗ ΤΟΥ ΜΟΝΤΕΛΟΥ

ΠΑΡΑΔΕΙΓΜΑ 1: One-way Anova

Έχουμε μία ποσοτική Y και μία κατηγορική A με k επίπεδα

$$Y_i \sim \text{Normal}(\mu_i, \sigma^2)$$

⌘ **Παραμετροποίηση 2:** $\mu_i = \mu + a_i$

⌘ $a_1 + a_2 + \dots + a_k = 0$ (STZ)

⌘ Ερμηνεία:

⌘ μ =συνολικό μέσος

⌘ a_i =διαφορά της ομάδας i από το συνολικό μέσο
(group effect)

5.2. ΠΑΡΑΜΕΤΡΟΠΟΙΗΣΗ ΤΟΥ ΜΟΝΤΕΛΟΥ

ΠΑΡΑΔΕΙΓΜΑ 1: One-way Anova

Έχουμε μία ποσοτική Y και μία κατηγορική A με k επίπεδα

$$Y_i \sim \text{Normal}(\mu_i, \sigma^2)$$

⌘ **Παραμετροποίηση 3:** $\mu_i = \mu + a_i$

⌘ $a_1 = 0$ (CR)

⌘ Ερμηνεία:

⌘ μ =μέσος μιάς ομάδας βάσης (baseline group)

συνήθως η ομάδα ελέγχου (control group)

a_i =διαφορά της ομάδας i από τη ομάδα βάσης

5.2. ΠΑΡΑΜΕΤΡΟΠΟΙΗΣΗ ΤΟΥ ΜΟΝΤΕΛΟΥ

ΨΕΥΔΟΜΕΤΑΒΛΗΤΕΣ: One-way Anova

- ⌘ **Παραμετροποίηση 3:** $\mu_i = \mu + a_i$
- ⌘ $a_1=0$ (CR)
- ⌘ $\mu = X \beta$, με $\beta^T = (\mu, a_2, \dots, a_k)$
- ⌘ $k-1$ dummies (D_2, D_3, \dots, D_k)
 - ◻ $D_j = 1$ αν $A=j$
 - ◻ $D_j = 0$ διαφορετικά

5.2. ΠΑΡΑΜΕΤΡΟΠΟΙΗΣΗ ΤΟΥ ΜΟΝΤΕΛΟΥ

ΨΕΥΔΟΜΕΤΑΒΛΗΤΕΣ: One-way Anova

- ⌘ **Παραμετροποίηση 2:** $\mu_i = \mu + a_i$
- ⌘ $a_1+a_2+\dots+a_k=0$ (STZ)
- ⌘ $\mu = X \beta$, με $\beta^T = (\mu, a_2, \dots, a_k)$
- ⌘ $k-1$ dummies (D_2, D_3, \dots, D_k)
 - ◻ $D_j = -1$ αν $A=1$
 - ◻ $D_j = 1$ αν $A=j$
 - ◻ $D_j = 0$ διαφορετικά

5.2. ΠΑΡΑΜΕΤΡΟΠΟΙΗΣΗ ΤΟΥ ΜΟΝΤΕΛΟΥ

ΨΕΥΔΟΜΕΤΑΒΛΗΤΕΣ: One-way Anova

για $k=3$

j	D₂	D₃	S₂	S₃
1	0	0	-1	-1
2	1	0	1	0
3	0	1	0	1

5.2. ΠΑΡΑΜΕΤΡΟΠΟΙΗΣΗ ΤΟΥ ΜΟΝΤΕΛΟΥ

ΣΧΕΣΕΙΣ ΠΑΡΑΜΕΤΡΩΝ: One-way Anova

$$\mu_i = \mu^{CR} + a_i^{CR} \Rightarrow$$

$$\mu_i = \mu^{CR} + a + a_i^{CR} - \bar{a} \Rightarrow$$

$$\mu_i = \mu^{STZ} + a_i^{STZ} \text{ με}$$

$$\mu^{STZ} = \mu^{CR} + a$$

$$a_i^{STZ} = a_i^{CR} - a$$

$$\text{Γενικά θέτουμε } \mathbf{X}^{STZ} \boldsymbol{\beta}^{STZ} = \mathbf{X}^{CR} \boldsymbol{\beta}^{CR} \Rightarrow$$

$$\boldsymbol{\beta}^{STZ} = (\mathbf{X}^{STZ})^{-1} \mathbf{X}^{CR} \boldsymbol{\beta}^{CR}$$

$$\boldsymbol{\beta}^{CR} = (\mathbf{X}^{CR})^{-1} \mathbf{X}^{STZ} \boldsymbol{\beta}^{STZ}$$

5.2. ΠΑΡΑΜΕΤΡΟΠΟΙΗΣΗ ΤΟΥ ΜΟΝΤΕΛΟΥ

ΠΑΡΑΔΕΙΓΜΑ 2: Απλή Παλινδρόμιση

Έχουμε μία ποσοτική Y και μία επεξηγηματική ποσοτική $X \Rightarrow Y_i \sim \text{Normal}(\mu_i, \sigma^2)$

⌘ **Παραμετροποίηση 1:** $\mu_i = \alpha + \beta X_i$

⌘ **Παραμετροποίηση 2:** $\mu_i = \alpha^* + \beta^*(X_i - \mu_x)$

⌘ **Παραμετροποίηση 3:** $\mu_i = \alpha^{**} + \beta^{**}(X_i - \mu_x)/\sigma_x$

5.2. ΠΑΡΑΜΕΤΡΟΠΟΙΗΣΗ ΤΟΥ ΜΟΝΤΕΛΟΥ

ΠΑΡΑΔΕΙΓΜΑ 2: Απλή Παλινδρόμιση

ΕΡΜΗΝΕΙΑ ΠΑΡΑΜΕΤΡΩΝ

⌘ **Παραμετροποίηση 1:**

⌘ α = Αναμενόμενος μέσος της Y όταν $X=0$

⌘ β = Αναμενόμενη αύξηση της Y όταν X αυξηθεί κατά μία μονάδα

5.2. ΠΑΡΑΜΕΤΡΟΠΟΙΗΣΗ ΤΟΥ ΜΟΝΤΕΛΟΥ

ΠΑΡΑΔΕΙΓΜΑ 2: Απλή Παλινδρόμιση ΕΡΜΗΝΕΙΑ ΠΑΡΑΜΕΤΡΩΝ

⌘ **Παραμετροποίηση 2:**

- ⌘ α^* = Αναμενόμενος μέσος της Y όταν $X = \mu_x$
- ⌘ β^* = Αναμενόμενη αύξηση της Y όταν X αυξηθεί κατά μία μονάδα

5.2. ΠΑΡΑΜΕΤΡΟΠΟΙΗΣΗ ΤΟΥ ΜΟΝΤΕΛΟΥ

ΠΑΡΑΔΕΙΓΜΑ 2: Απλή Παλινδρόμιση ΕΡΜΗΝΕΙΑ ΠΑΡΑΜΕΤΡΩΝ

⌘ **Παραμετροποίηση 3:**

- ⌘ α^{**} = Αναμενόμενος μέσος του Y όταν $X = \mu_x$
- ⌘ β^{**} = Αναμενόμενη αύξηση της Y όταν X αυξηθεί κατά ποσότητα ίση με την τυπική απόκλιση του X [είναι ίσος με το συντελεστή συσχέτισης, λέγεται και beta parameter]

5.2. ΠΑΡΑΜΕΤΡΟΠΟΙΗΣΗ ΤΟΥ ΜΟΝΤΕΛΟΥ

ΠΑΡΑΔΕΙΓΜΑ 2: Απλή Παλινδρόμιση ΣΧΕΣΕΙΣ ΠΑΡΑΜΕΤΡΩΝ

Παραμετροποίηση 2:

$$\textcircled{2} \mu_i = \alpha^* + \beta^*(X_i - \mu_x) = \alpha^* - \beta^* \mu_x + \beta^* X_i$$

- $\alpha = \alpha^* - \beta^* \mu_x$
- $\beta = \beta^*$

Παραμετροποίηση 3:

$$\textcircled{3} \mu_i = \alpha^{**} + \beta^{**}(X_i - \mu_x)/\sigma_x = \alpha^{**} - \beta^{**} \mu_x/\sigma_x + \beta^{**} X_i/\sigma_x$$

- $\alpha = \alpha^{**} - \beta^{**} \mu_x/\sigma_x$
- $\beta = \beta^{**}/\sigma_x$

5.2. ΠΑΡΑΜΕΤΡΟΠΟΙΗΣΗ ΤΟΥ ΜΟΝΤΕΛΟΥ

Διαφορετική προσδιορίσιμη παραμετροποίηση **δεν επιδρά** στις posterior

Διαφορετική προσδιορίσιμη παραμετροποίηση **επιδρά** στη σύγκλιση του MCMC [πρεπει οι παράμετροι να είναι όσο το δυνατόν λιγότερο συσχετισμένοι]

Για τις μη-προσδιορίσιμες παράμετρους ενός μοντέλου

- 1... MCMC -> δε συγκλίνει
- 2... Η posterior θα είναι improper
- 3... Συναρτήσεις αυτών μπορει να συγκλίνουν!?

5.3. RANDOM EFFECTS

Τι είναι τα random effects

- ⌘ Τυχαίο σφάλμα στον γραμμικό προσδιορισμό (η)
- ⌘ Οι παράμετροι (στην κλασσική στατιστική) δεν είναι σταθερές αλλά τυχαίες μεταβλητές

5.3. RANDOM EFFECTS

Πότε χρησιμοποιούμε random effects

- ⌘ Επαναλαμβανόμενες μετρήσεις στα ίδια άτομα την ίδια χρονική περίοδο (διασπορά λόγω σφάματος μέτρησης ή τυχαιότητας του ίδιου παράγοντα)
- ⌘ Επαναλαμβανόμενες μετρήσεις σε διαφορετικές χρονικές περιόδους
- ⌘ Χρήση πιο σύνθετων κατανομών (Mixtures)
- ⌘ Διόρθωση Υπερ-διασποράς (overdispersion)
- ⌘ Εκτίμηση της within-subject variability

Bayesian Biostatistics Using BUGS



5.4. ΠΑΡΑΔΕΙΓΜΑΤΑ ΣΤΟ BUGS

Department of Statistics,
Athens University of
Economics & Business

5.4.1. Normal Hierarchical Model with Missing Data (Rats)

BUGS Examples Vol.1, σελ.4, Example 1

- # Section 6 of Gelfand *et al* (1990, JASA),
- # 30 young rats
- # weights for 5 weeks.


5.4.1. Normal Hierarchical Model with Missing Data (Rats)

Part of the data:

	Weights Y_{ij} of rat i on day x_j				
	$x_j = 8$	15	22	29	36
Rat 1	151	199	246	283	320
Rat 2	145	199	249	293	354
...
Rat 30	153	200	244	286	324

Y_{ij} is the weight of the i th rat measured at age x_j .

5.4.1. Normal Hierarchical Model with Missing Data (Rats)



5.4.1. Normal Hierarchical Model with Missing Data (Rats)

The model is essentially a random effects linear growth curve

$$Y_{ij} \sim \text{Normal}(\alpha_i + \beta_i(x_i - \bar{x}), \tau_c)$$

$$\alpha_i \sim \text{Normal}(\alpha_c, \tau_\alpha)$$

$$\beta_i \sim \text{Normal}(\beta_c, \tau_\beta)$$

where $\bar{x} = 22$, and τ represents the *precision* (1/variance) of a normal distribution.

see the `Birats` example in Volume 2 which models the covariance between α_i and β_i .

standardise the x_i 's around their mean to reduce dependence
(for the full balanced data, complete independence is achieved)

5.4.1. Normal Hierarchical Model with Missing Data (Rats)

a_c, t_a, b_c, t_b, t_c are given independent "noninformative" priors.


Interest particularly focuses on the intercept at zero time (birth), denoted $a_0 = a_c - b_c \bar{x}$.

RANDOM EFFECTS => two sources of variability

- 1... within each subject (rat)
- 2... across subjects

5.4.1. Normal Hierarchical Model with Missing Data (Rats)

Graphical model for rats example:



5.4.1. Normal Hierarchical Model with Missing Data (Rats)


BUGS language for rats example:

```

model
{
  for(i in 1:N){
    for(j in 1:T){
      Y[i, j] ~ dnorm(mu[i, j], tau.c)
      mu[i, j] <- alpha[i] + beta[i] * (x[j] - xbar)
    }
    alpha[i] ~ dnorm(alpha.c, alpha.tau)
    beta[i] ~ dnorm(beta.c, beta.tau)
  }
  tau.c ~ dgamma(0.001, 0.001)
  sigma <- 1 / sqrt(tau.c)
  alpha.c ~ dnorm(0.0, 1.0E-6)
  alpha.tau ~ dgamma(0.001, 0.001)
  beta.c ~ dnorm(0.0, 1.0E-6)
  beta.tau ~ dgamma(0.001, 0.001)
  alpha0 <- alpha.c - xbar * beta.c
}

```

5.4.1. Normal Hierarchical Model with Missing Data (Rats)



5.4.1. Normal Hierarchical Model with Missing Data (Rats)

node	mean	sd	MC error	2.5%	median	97.5%	start	sample
alpha0	106.4	3.518	0.1071	99.87	106.4	113.7	1001	1000
beta.c	6.187	0.1052	0.003881	5.978	6.189	6.396	1001	1000
sigma	6.123	0.5053	0.02685	5.288	6.065	7.231	1001	1000

5.4.1. Normal Hierarchical Model with Missing Data (Rats)

MISSING VALUES ANALYSIS(ratmiss)

Missing Value -> NA

Rat 26 -> 153 NA NA NA NA

Monitor (Y[26,]) -> predicted values of Missing

node	mean	sd	MC error	2.5%	median	97.5%
Y[26,2]	205.3	8.711	0.3794	187.6	205.2	222.3
Y[26,3]	251.5	10.39	0.5242	230.7	251.4	272.7
Y[26,4]	297.1	12.54	0.6756	271.1	297.4	321.0
Y[26,5]	343.3	15.55	0.8251	313.7	343.6	373.6

5.4.1. Normal Hierarchical Model with Missing Data (Rats)

FULL DATA ANALYSIS

node	mean	sd	MC error	2.5%	median	97.5%	start	sample
alpha0	106.4	3.518	0.1071	99.87	106.4	113.7	1001	1000
beta.c	6.187	0.1052	0.003881	5.978	6.189	6.396	1001	1000
sigma	6.123	0.5053	0.02685	5.288	6.065	7.231	1001	1000

MISSING DATA ANALYSIS

node	mean	sd	MC error	2.5%	median	97.5%	start	sample
alpha0	100.9	3.75	0.1106	93.63	101.0	108.3	1001	2000
beta.c	6.581	0.1449	0.00655	6.298	6.583	6.872	1001	2000
sigma	6.129	0.7062	0.03879	4.93	6.065	7.728	1001	2000

5.4.2. Random Effects Logistic Regression (Seeds)

BUGS Examples Vol.1, σελ.10, Example 3

- ⌘ Table 3 of Crowder (1978, *Ap.Stat.*), also analysed by Breslow & Clayton (1993, *Ap.Stat.*)
- ⌘ % of germinated seeds on 21 plates
- ⌘ 2 by 2 factorial layout by seed and type of root extract
- ⌘ r_i = # germinated seeds in i plate
- ⌘ n_i = total # seeds in i plate
- ⌘ $i=1, \dots, N=21$ plates

5.4.2. Random Effects Logistic Regression (Seeds)

seed O. aegyptiaco 75			seed O. aegyptiaco 73								
Bean			Cucumber			Bean			Cucumber		
r	n	r/n	r	n	r/n	r	n	r/n	r	n	r/n
10	39	0.26	5	6	0.83	8	16	0.50	3	12	0.25
23	62	0.37	53	74	0.72	10	30	0.33	22	41	0.54
23	81	0.28	55	72	0.76	8	28	0.29	15	30	0.50
26	51	0.51	32	51	0.63	23	45	0.51	32	51	0.63
17	39	0.44	46	79	0.58	0	4	0.00	3	7	0.43
			10	13	0.77						

5.4.2. Random Effects Logistic Regression (Seeds)

DATA IN BUGS

```
r[] n[] x1[] x2[]
10 39 0 0
...
5 6 0 1
...
8 16 1 0
...
3 12 1 1
x1 = dummy for seed type
x2 = dummy for root extract
```

5.4.2. Random Effects Logistic Regression (Seeds)

MODEL

1.. RANDOM COMPONENT:

$$\# r_i \sim \text{Binomial}(p_i, n_i)$$


2.. Systematic Component & link:

$$\# \text{logit}(p_i) = \alpha_0 + \alpha_1 x_{1i} + \alpha_2 x_{2i} + \alpha_{12} x_{1i} x_{2i} + b_i$$

3.. Random Effect:

$$\# b_i \sim \text{Normal}(0, \tau)$$

5.4.2. Random Effects Logistic Regression (Seeds)



5.4.2. Random Effects Logistic Regression (Seeds)

```

model
{
  # Model Likelihood
  for( i in 1 : N ) {
    r[i] ~ dbin(p[i],n[i]) # random component
    b[i] ~ dnorm(0.0,tau)  # random effect
    #systematic component
    logit(p[i]) <- alpha0 + alpha1 * x1[i] +
      alpha2 * x2[i] + alpha12 * x1[i] * x2[i] +
      b[i]
  }
}

```

5.4.2. Random Effects Logistic Regression (Seeds)

```
...
# Prior distributions
alpha0 ~ dnorm(0.0,1.0E-6)
alpha1 ~ dnorm(0.0,1.0E-6)
alpha2 ~ dnorm(0.0,1.0E-6)
alpha12 ~ dnorm(0.0,1.0E-6)
tau ~ dgamma(0.001,0.001)
sigma <- 1 / sqrt(tau)
}
```

5.4.2. Random Effects Logistic Regression (Seeds)

node	mean	sd	MC error	2.5%	median	97.5%	start	sample
alpha0	-0.5546	0.1941	0.007696	-0.9353	-0.5577	-0.1597	1001	10000
alpha1	0.08497	0.3127	0.01283	-0.5814	0.09742	0.6679	1001	10000
alpha12	-0.8229	0.4321	0.01785	-1.697	-0.8218	0.01641	1001	10000
alpha2	1.356	0.2743	0.01236	0.8257	1.347	1.909	1001	10000
siama	0.2731	0.1437	0.007956	0.04133	0.2654	0.5862	1001	10000

We may compare simple logistic, maximum likelihood (from EGRET), penalized quasi-likelihood (PQL) Breslow and Clayton (1993) with the BUGS results

variable	Logistic regression		maximum likelihood		PQL	
	β	SE	β	SE	β	SE
α_0	-0.558	0.126	-0.546	0.167	-0.542	0.190
α_1	0.146	0.223	0.097	0.278	0.77	0.308
α_2	1.318	0.177	1.337	0.237	1.339	0.270
α_{12}	-0.778	0.306	-0.811	0.385	-0.825	0.430
σ	---	---	0.236	0.110	0.313	0.121

5.4.2. Random Effects Logistic Regression (Seeds)

CONSTRAINING RANDOM EFFECTS

- 1... Sum-to-zero constraints
- 2... Hierarchical Centering

ALTERNATIVE PARAMETRIZATION FOR Sigma

5.4.2. Random Effects Logistic Regression (Seeds)

CONSTRAINING RANDOM EFFECTS

- 1... Sum-to-zero constraints

Why:

- (a) Unconstrained b_i may not converge
- (b) Estimate fixed effects for each subject

Use $c[i]$:

```
c[i] ~ dnorm(0, tau)  
b[i] <- c[i] - mean(c[])
```

5.4.2. Random Effects Logistic Regression (Seeds)

CONSTRAINING RANDOM EFFECTS

2... Hierarchical Centering


$$\mu_i = \alpha_0 + \alpha_1 x_{1i} + \alpha_2 x_{2i} + \alpha_{12} x_{1i} x_{2i}$$

$$\beta_i = \mu_i + b_i$$


$$\text{logit}(p_i) = \beta_i$$

$$\beta_i \sim \text{Normal}(\mu_i, \tau)$$

5.4.2. Random Effects Logistic Regression (Seeds)



5.4.2. Random Effects Logistic Regression (Seeds)



5.4.2. Random Effects Logistic Regression (Seeds)

CONSTRAINING RANDOM EFFECTS

2... Hierarchical Centering

advantages:

- (a) Gibbs sampler with better correlation properties
- (b) Conditionals for the α parameters are conjugate=> reduced generation time

5.4.2. Random Effects Logistic Regression (Seeds)

ALTERNATIVE PARAMETRIZATION FOR Sigma

```
b[i] ~ dnorm(0.0,1.0)  # random effect  
#systematic component  
logit(p[i]) <- alpha0 + alpha1 * x1[i] +  
alpha2 * x2[i] + alpha12 * x1[i] * x2[i] +  
sigma*b[i]  
# exponential prior on sigma  
sigma~dexp(1.0)
```

5.4.3. Institutional Ranking (Surgical)

BUGS Examples Vol.1, σελ.15, Example 4

Mortality rates in 12 hospitals performing cardiac surgery in babies.

5.4.3. Institutional Ranking (Surgical)

Hospital	No of ops	No of deaths
A	47	0
B	148	18
C	119	8
D	810	46
E	211	8
F	196	13
G	148	9
H	215	31
I	207	14
J	97	8
K	256	29
L	360	24

5.4.3. Institutional Ranking (Surgical)

MODEL

$$r_i \sim \text{Binomial}(p_i, n_i)$$

(1) **Fixed Effects Model:** $p_i \sim \text{Beta}(1.0, 1.0)$

[Uniform Prior on each p_i]

(2) **Random Effects Model:**

$$\text{logit}(p_i) = b_i$$

$$b_i \sim \text{Normal}(\mu, \tau)$$

$$\mu \sim \text{Normal}(0.0, 10^{-6}) \text{ [Prior]}$$

$$\tau \sim \text{Gamma}(10^{-6}, 10^{-6}) \text{ [Prior]}$$

5.4.3. Institutional Ranking (Surgical)

BUGS CODE: FIXED EFFECTS MODEL

```
model{
  for( i in 1 : N ) {

    r[i] ~ dbin(p[i], n[i])


    p[i] ~ dbeta(1.0, 1.0)
  }
}
```

5.4.3. Institutional Ranking (Surgical)


BUGS CODE: RANDOM EFFECTS MODEL

```
model{
  for( i in 1 : N ) {
    b[i] ~ dnorm(mu,tau)
    r[i] ~ dbin(p[i],n[i])
    logit(p[i]) <- b[i]
  }
  pop.mean <- exp(mu) / (1 + exp(mu))
  mu ~ dnorm(0.0,1.0E-6)
  sigma <- 1 / sqrt(tau)
  tau ~ dgamma(0.001,0.001)
}
```

5.4.3. Institutional Ranking (Surgical)



5.4.3. Institutional Ranking (Surgical)




5.4.3. Institutional Ranking (Surgical)

RANKING HOSPITALS (WINBUGS)

- 1... Update 1000 [burnin]
- 2... Select ***Inference>Rank***
- 3... Set the rank monitor for p
- 4... Update 10,000 iterations
- 5... Select "histogram" or "Stats" option.

5.4.3. Institutional Ranking (Surgical)


RANKING HOSPITALS (WINBUGS)



5.4.3. Institutional Ranking (Surgical)

RANKING HOSPITALS (WINBUGS)


histogram




Ranks

node	2.5%	median	97.5%
p[1]	1	2	10
p[2]	6	10	12
p[3]	1	6	11
p[4]	1	4	8
p[5]	1	2	8
p[6]	1	6	10
p[7]	1	5	10
p[8]	9	12	12
p[9]	1	6	10
p[10]	2	8	12
p[11]	6	10	12
p[12]	1	6	10

5.4.3. Institutional Ranking (Surgical)



5.4.3. Institutional Ranking (Surgical)



5.4.3. Institutional Ranking (Surgical)

RESULTS

Considerable uncertainty associated with 'league tables':

- (a) only 2 hospitals (H and K) exclude the median rank
- (b) none of intervals fall completely within the lower or upper quartiles

5.4.3. Institutional Ranking (Surgical)

RANKING HOSPITALS (Classic BUGS)

Compute Ranks in Bugs:

```
for (i in 1:N) {  
    for (j in 1:N) {  
        not.less.than[i,j]<-step(p[i]-p[j]);  
    }  
    rank[i]<-sum(not.less.than[i,]);  
}  
step(x)=1 if x≥0 and step(x)=0 if x<0  
monitor(rank) # monitor ranks in BUGS  
stats(rank) # view stats of ranks in BUGS
```

5.4.4. Cross Over Trial (equiv)

BUGS Examples Vol.1, σελ. 21, Example 6

Data from a two-treatment, two-period crossover trial to compare 2 tablets A and B, as reported by Gelfand *et al* (1990, JASA).

5.4.4. Cross Over Trial (equiv)

Data

Subject i	Sequence	seq	Period 1	T_{i1}	Period 2	T_{i2}
1	AB	1	1.40	1	1.65	2
2	AB	1	1.64	1	1.57	2
3	BA	-1	1.44	2	1.58	1
...						
8	AB	1	1.25	1	1.44	2
9	BA	-1	1.25	2	1.39	1
10	BA	-1	1.30	2	1.52	1


5.4.4. Cross Over Trial (equiv)

Definitions

Two-treatment, Two-period Cross-over trial

- ⌘ Type of randomized clinical trial
- ⌘ Each subject is randomly allocated in two groups
- ⌘ Group 1: receive drug A in period 1 and drug B in period 2
- ⌘ Group 2: receive drug B in period 1 and drug A in period 2
- ⌘ *Wash-out Period*: period of no-treatment between two active drug periods

5.4.4. Cross Over Trial (equiv)



5.4.4. Cross Over Trial (equiv)

Definitions

Treatment or Drug Effect: The effect of different treatment or medication

Carry Over Effect: residual biological effect of a 1st period study treatment
Wash-out period often does not eliminates carry-over effects

5.4.4. Cross Over Trial (equiv)

Classical Inference for Treatment Effects

Y_{ik} measurement of i subject, k period

Find the differences [Drug 1 -Drug 2]

$$\Delta_{i1} = D_{i1} = Y_{i1} - Y_{i2} \text{ for group 1 } [T_1 - T_2 + P_1 - P_2]$$

$$\Delta_{i2} = -D_{i2} = Y_{i2} - Y_{i1} \text{ for group 2 } [T_1 - T_2 + P_2 - P_1]$$

Estimate overall effect by

$$\bar{\Delta} = (\bar{\Delta}_1 + \bar{\Delta}_2) / 2 = (\bar{D}_1 - \bar{D}_2) / 2$$

Test H_0 : Drug-effect [$E(\Delta)$] = 0 using one-sample t-test

5.4.4. Cross Over Trial (equiv)

Classical Inference for Carry-over Effects

Previous analysis assumed no carry-over effects

Find the sum [Drug 1 +Drug 2]: $S_{ij} = Y_{i1} + Y_{i2}$

Estimate overall carry-over effect by $\bar{S} = \bar{S}_1 - \bar{S}_2$

Test H_0 : Carry-over effect = 0 using two-sample t-test

5.4.4. Cross Over Trial (equiv)

MODEL BUILDING

Find the differences [Drug 1 -Drug 2]

$$Y_{i1} - Y_{i2} \sim \text{Normal} (\mu_1, \tau^2) \text{ via Group 1 [AB]}$$

$$Y_{i1} - Y_{i2} \sim \text{Normal} (\mu_2, \tau^2) \text{ via Group 2 [BA]}$$

$$\mu_1 = \pi_1 - \pi_2 + d_1 - d_2 \text{ and } \mu_2 = \pi_1 - \pi_2 - d_1 + d_2$$

$$\textbf{Drug effect } \varphi = (\mu_2 - \mu_1)/2 = (d_2 - d_1)/2$$

$$\textbf{Period Effect } \pi = (\mu_1 + \mu_2)/2 = (\pi_1 - \pi_2)/2$$

[if $\pi_1 = \pi_2$ no period/carry over effect]

$$\text{Reparametrize } \mu_1 = (\pi - \varphi), \mu_2 = (\pi + \varphi)/2$$

therefore $D_i = Y_{i1} - Y_{i2} \sim \text{Normal} ((\pi + \text{seq } \varphi)/2, \tau^2)$
 seq=-1 if group AB, 1 if group BA

5.4.4. Cross Over Trial (equiv)

MODEL BUILDING

$$Y_{i1} + Y_{i2} \sim \text{Normal} (m_1, \tau^{*2}) \text{ via Group 1 [AB]}$$

$$Y_{i1} + Y_{i2} \sim \text{Normal} (m_2, \tau^{*2}) \text{ via Group 2 [BA]}$$

$$\text{Carry Over effect cof} = m_1 - m_2$$

If cof=0 ($m_1 = m_2$) no carry-over effect

$$\text{Reparametrize } m_1 = \mu - \text{cof}, m_2 = \mu + \text{cof}$$

5.4.4. Cross Over Trial (equiv)

Random Effects model (No carry-over effect)

$$\begin{aligned} Y_{ik} &\sim \text{Normal}(m_{ik}, t1) \\ m_{ik} &= m + (-1)^{Tik-1} \varphi/2 + (-1)^{k-1}\pi/2 + d_i \\ d_i &\sim \text{Normal}(0, t2) \end{aligned}$$

5.4.4. Cross Over Trial (equiv)

Random Effects model (BUGS CODE)

```
model{
  for( k in 1 : P ) {
    for( i in 1 : N ) {
      Y[i,k] ~ dnorm(m[i,k],tau1)
      m[i,k] <- mu + sign[T[i,k]] * phi / 2 +
        sign[k] * pi / 2 + delta[i]
      T[i,k] <- group[i] * (k - 1.5) + 1.5}
    for( i in 1 : N ) {delta[i] ~ dnorm(0.0, tau2)}
```

5.4.4. Cross Over Trial (equiv)

```
Random Effects model (BUGS CODE)
tau1 ~ dgamma(0.001, 0.001)
sigma1 <- 1 / sqrt(tau1)
tau2 ~ dgamma(0.001, 0.001)
sigma2 <- 1 / sqrt(tau2)
mu ~ dnorm(0.0, 1.0E-6)
phi ~ dnorm(0.0, 1.0E-6)
pi ~ dnorm(0.0, 1.0E-6)
theta <- exp(phi)
equiv <- step(theta - 0.8) - step(theta - 1.2)
}
```

5.4.4. Cross Over Trial (equiv)

	Period 1	Period 2
AB	$\mu + \varphi/2 + \pi/2$	$\mu - \varphi/2 - \pi/2$
BA	$\mu - \varphi/2 + \pi/2$	$\mu + \varphi/2 - \pi/2$

$$\text{cof} = [(\mu + \varphi/2 + \pi/2) + (\mu - \varphi/2 - \pi/2)] \\ - [(\mu - \varphi/2 + \pi/2) + (\mu + \varphi/2 - \pi/2)] = 0$$

5.4.4. Cross Over Trial (equiv)

Random Effects model (carry-over effect)

$$\begin{aligned}
 Y_{ik} &\sim \text{Normal}(m_{ik}, t1) \\
 m_{ik} &= \mu + (-1)^{T_{ik}-1} \varphi/2 + (-1)^{k-1} \pi/2 + d_i \\
 &\quad + (k-1) (-1)^{T_{ik}} \text{cof}/2 \\
 d_i &\sim \text{Normal}(0, t2)
 \end{aligned}$$

5.4.4. Cross Over Trial (equiv)

	Period 1	Period 2
AB	$\mu + \varphi/2 + \pi/2$	$\mu - \varphi/2 - \pi/2 + \text{cof}/2$
BA	$\mu - \varphi/2 + \pi/2$	$\mu + \varphi/2 - \pi/2 - \text{cof}/2$

$$\begin{aligned}
 \text{cof} &= [(\mu + \varphi/2 + \pi/2) + (\mu - \varphi/2 - \pi/2 + \text{cof}/2)] \\
 &\quad - [(\mu - \varphi/2 + \pi/2) + (\mu + \varphi/2 - \pi/2 - \text{cof}/2)]
 \end{aligned}$$

5.4.5. Variance Components Models (Dyes)

BUGS Examples Vol.1, σελ. 24, Example 7

Box and Tiao (1973) analyse data first presented by Davies (1967) concerning batch to batch variation in yields of dyestuff.

The data (shown below) arise from a balanced experiment whereby the total product yield was determined for 5 samples from each of 6 randomly chosen batches of raw material.

5.4.5. Variance Components Models (Dyes)

Batch	Yield (in grams)				
1	1545	1440	1440	1520	1580
2	1540	1555	1490	1560	1495
3	1595	1550	1605	1510	1560
4	1445	1440	1595	1465	1545
5	1595	1630	1515	1635	1625
6	1520	1455	1450	1480	1445

5.4.5. Variance Components Models (Dyes)

Object of the study:

determine the relative importance of ***between batch variation*** versus ***variation due to sampling and analytic errors.***

Assumption:

batches and samples vary independently, and contribute additively to the total error variance

5.4.5. Variance Components Models (Dyes)

MODEL

$$y_{ij} \sim \text{Normal}(m_i, t_{\text{within}})$$

$$m_i \sim \text{Normal}(q, t_{\text{between}})$$

y_{ij} =yield for sample j of batch i ,

m_i =true yield for batch i ,

t_{within} is the inverse of the within-batch variance

s_{within}^2 variation due to sampling and analytic error

q true average yield for all batches

t_{between} inverse of the between-batch variance

s_{between}^2 .

5.4.5. Variance Components Models (Dyes)

MODEL

Total variation $\sigma_{\text{total}}^2 = \sigma_{\text{within}}^2 + \sigma_{\text{between}}^2$

Relative contributions of each component to the total variance are


$$\begin{aligned} f_{\text{within}} &= \sigma_{\text{within}}^2 / \sigma_{\text{total}}^2 \text{ and} \\ f_{\text{between}} &= \sigma_{\text{between}}^2 / \sigma_{\text{total}}^2. \end{aligned}$$

5.4.5. Variance Components Models (Dyes)

BUGS CODE

```
model{
  for( i in 1 : batches ) {
    for( j in 1 : samples ) {
      y[i , j] ~ dnorm(m[i], tau.with)
      m[i] ~ dnorm(theta, tau.btw)
    }
    sigma2.with <- 1 / tau.with
    sigma2.btw <- 1 / tau.btw
    sigma2.total<- sigma2.btw+ sigma2.with
    f.with<- sigma2.with/ sigma2.total
    f.btw<- sigma2.btw/ sigma2.total
    tau.with ~ dgamma(0.001, 0.001)
    tau.btw ~ dgamma(0.001, 0.001)
    theta ~ dnorm(0.0, 1.0E-10)
  }
}
```

5.4.5. Variance Components Models (Dyes)



5.4.5. Variance Components Models (Dyes)

node	mean	sd	MC error	2.5%	median	97.5%
<code>f.btw</code>	0.3311	0.2463	0.006816	1.503E-6	0.3261	0.8137
<code>f.with</code>	0.6689	0.2463	0.006816	0.1863	0.6739	1.0
<code>sigma2.btw</code>	2194.0	3832.0	49.2	0.006255	1277.0	10230.0
<code>sigma2.with</code>	3059.0	1138.0	25.65	1562.0	2815.0	5880.0
<code>theta</code>	1527.0	21.84	0.2532	1485.0	1527.0	1571.0

5.4.6. Random Effects Meta-analysis of Clinical Trials(Blocker)

BUGS Examples Vol.1, σελ. 33, Example 10

Carlin (1992, Stats in Medicine)

22 trials of beta-blockers to prevent mortality after myocardial infarction

Study	Mortality: deaths / total	
	Treated	Control
1	3/38	3/39
2	7/114	14/116
.....		
22	22/680	39/674

5.4.6. Random Effects Meta-analysis of Clinical Trials(Blocker)

Random effects meta-analysis:

Assumption:

the true effect (on a log-odds scale) δ_i in a trial i is drawn from some population distribution.

r_{C_i} number of events in the control group in trial i ,

r_{T_i} number of events under active treatment in trial i .

MODEL

$$r_{C_i} \sim \text{Binomial}(p_{C_i}, n_{C_i})$$

$$r_{T_i} \sim \text{Binomial}(p_{T_i}, n_{T_i})$$

$$\text{logit}(p_{C_i}) = \mu_i$$

$$\text{logit}(p_{T_i}) = \mu_i + \delta_i$$

$$\delta_i \sim \text{Normal}(d, \tau)$$

5.4.6. Random Effects Meta-analysis of Clinical Trials(Blocker)

BUGS CODE

```

model{
  for( i in 1 : Num ) {
    rc[i] ~ dbin(pc[i], nc[i])
    rt[i] ~ dbin(pt[i], nt[i])
    logit(pc[i]) <- mu[i]
    logit(pt[i]) <- mu[i] + delta[i]
    mu[i] ~ dnorm(0.0,1.0E-5)
    delta[i] ~ dnorm(d, tau)    }
  d ~ dnorm(0.0,1.0E-6)
  tau ~ dgamma(0.001,0.001)
  delta.new ~ dnorm(d, tau)
  sigma <- 1 / sqrt(tau)      }

```

5.4.7. Weibull Regression in Censored Survival Analysis

BUGS Examples Vol.1, σελ. 54, Example 16

Dellaportas and Smith (1993, Ap. Stats) analyse data from Grieve (1987, Statistician)

Photocarcinogenicity in four groups, each containing 20 mice.

Recorded:

survival time

whether they died or were censored at that time.

A portion of the data, giving survival times in weeks, are shown below. A * indicates censoring.

5.4.7. Weibull Regression in Censored Survival Analysis

DATA

Mouse	Irradiated control	Vehicle substance	Test control	Positive
1 12	32	22	27	
.....				
18 *40	30	24	12	
19 31	37	37	17	
20 36	27	29	26	

* indicates censoring.

5.4.7. Weibull Regression in Censored Survival Analysis

The survival distribution is assumed to be Weibull.

$$t_i \sim \text{Weibull}(\tau, \mu_i)$$

$$\mu_i = \exp(\beta z_i)$$

$$f(t_i, z_i) = r \mu_i t_i^{r-1} \exp(-\mu_i \exp(t_i^r))$$

t_i failure time of an individual with covariate vector z_i and β is a vector of unknown regression coefficients.

$$\text{Baseline hazard function: } \lambda_0(t_i) = rt_i^{r-1}$$

For censored observations, the survival distribution is a truncated Weibull, with lower bound corresponding to the censoring time.

5.4.7. Weibull Regression in Censored Survival Analysis

Prior distributions

$$f(\beta) = \text{Normal}(0, \tau = 0.0001)$$

$$f(r) = \text{Gamma}(1, 0.0001)$$

Median survival for individuals with covariate vector \mathbf{z}_i
 $m_i = (\exp(-\beta z_i) \log 2)^{1/r}$

5.4.7. Weibull Regression in Censored Survival Analysis

BUGS CODE

```

model{
  for(i in 1 : M) {   for(j in 1 : N) {
    t[i, j] ~ dweib(r, mu[i])I(t.cen[i, j],)
  }
  mu[i] <- exp(beta[i])
  beta[i] ~ dnorm(0.0, 0.001)
  median[i]<-pow(log(2)*exp(-beta[i]),1/r)
}
r ~ dexp(0.001)
veh.control <- beta[2] - beta[1]
test.sub <- beta[3] - beta[1]
pos.control <- beta[4] - beta[1]
}
```


5.4.7. Weibull Regression in Censored Survival Analysis

```

y[i, j] is observed time
ind[i,j] is censoring indicator (0=survival,
1=censored)
if i,j observation is survival time then
  t[i,j]=y[i,j]
  t.cen[i, j]=0
  dweib(r, mu[i])I(0,) = dweib(r, mu[i])
if i,j observation is censored observation
then
  t[i,j]=NA
  t.cen[i, j]=censored time = ind[i,j]*y[i,j]
  dweib(r, mu[i])I(t.cen[i, j],)

```

5.4.7. Weibull Regression in Censored Survival Analysis



5.4.7. Weibull Regression in Censored Survival Analysis

Other Survival Examples:

BUGS Examples Vol.1, σελ. 57, Example 17
(Weibull Regression with Random Effects, kidney)

BUGS Examples Vol.1, σελ. 60, Example 18
(Survival Analysis Using Cox Regresion, Leuk)

5.4.8. Multinomial Logistic Regression (Alli/Alligator)

BUGS Examples Vol.2, σελ. 51, Example 14

Agresti(1990, *Categorical Data Analysis*)

⌘ 4x2x5 Contingency Table

⌘ 221 alligators cross classified by

- ─ food type (fish, invertebrate, reptile, bird, other)
[Y]

- ─ length of alligator (≤ 2.3 and > 2.3)

- ─ lake (Hancock, Oklawaha, Trafford, George)

5.4.8. Multinomial Logistic Regression (Alli/Alligator)

Lake	Size	Primary Food Choice				
		Fish	Inv.	Reptile	Bird	Other
Hancock	<=2.3	23	4	2	2	8
	> 2.3	7	0	1	3	5
Olkawaha	<=2.3	5	11	1	0	3
	> 2.3	13	8	6	1	0
Trafford	<=2.3	5	11	2	1	5
	> 2.3	8	7	6	3	5
George	<=2.3	16	19	1	2	3
	> 2.3	17	1	0	1	3

5.4.8. Multinomial Logistic Regression (Alli/Alligator)

Assume multinomial response with a logit link.

$$X_{ij.} \sim \text{Multinomial}(p_{ij.}, n_{ij})$$

$$p_{ijk} = \phi_{ijk} / \sum_k \phi_{ijk}$$

$$\phi_{ijk} = \exp(\alpha_k + \beta_{ik} + \gamma_{jk})$$

$$X_{ij.} = X_{ij1}, \dots, X_{ij5}$$

$$n_{ij} = \sum_k X_{ijk}$$

$$\alpha_1 = \beta_{i1} = \gamma_{j1} = \gamma_{1k} = 0 \text{ (corner constraints)}$$

5.4.8. Multinomial Logistic Regression (Alli/Alligator)

BUGS CODE

```
...
# LIKELIHOOD
for (i in 1 : I) { for (j in 1 : J) {
  X[i,j,1:K] ~ dmulti( p[i,j,1:K] , n[i,j])
  n[i,j] <- sum(X[i,j,])
  for (k in 1:K) {
    p[i,j,k]<- phi[i,j,k]/sum(phi[i,j,])
    log(phi[i,j,k]) <- alpha[k] + beta[i,k]
    + gamma[j,k]
  }
}
```

5.4.8. Multinomial Logistic Regression (Alli/Alligator)

Alternatively assume Poisson response

$$X_{ijk} \sim \text{Poisson}(\mu_{ijk})$$

$$\log(\mu_{ijk}) = \lambda_{ij} + \alpha_k + \beta_{ik} + \gamma_{jk}$$

```
# POISSON LIKELIHOOD
for (i in 1 : I) { for (j in 1 : J) {
  for (k in 1:K){
    X[i,j,k] ~ dpois( mu[i,j,k])
    log(mu[i,j,k]) <- lambda[i,j] + alpha[k] +
      beta[i,k] + gamma[j,k]
  }
}}
```

5.4.9. Other Interesting Examples

BUGS Examples Vol.1, σελ. 30, Example 9
(Epil: repeated measures on Poisson Counts)

BUGS Examples Vol.2, σελ. 18, Example 7
(Jaw: repeated measures ANOVA)

BUGS Examples Vol.2, σελ. 55, Example 15
(Endo: conditional inference for matched samples in case-control studies)

Bayesian Biostatistics Using BUGS



ΤΕΛΟΣ Ζου ΜΑΘΗΜΑΤΟΣ

ΕΠΟΜΕΝΟ:
**MODEL DIAGNOSTICS AND
VARIABLE SELECTION**

E-mail: ntzoufras@aueb.gr

Department of Statistics,
Athens University of
Economics & Business